ΠΡΟΓΡΑΜΜΑ  ΠΛΑΙΥΩΧ

ΜΕΤΑΒΛΗΤΕΣ

  ΧΑΡΑΚΤΗΡΕΣ: ΑΓΩΝΕΣ[8,8], ΟΝ[8], ΟΜΙΛΟΣ[8]

  ΑΚΕΡΑΙΕΣ: ΒΑΘΜΟΙΠΡΩΤ[8], ΒΑΘΜΟΙ[8], ΣΟΥΜΑ[8], ΔΙΑΦ1, ΔΙΑΦ2, Ι, Κ, ΔΕΙΚΤΗΣ

  ΛΟΓΙΚΕΣ: ΦΡΕΝΟ

ΑΡΧΗ

  ΓΙΑ Ι ΑΠΟ 1 ΜΕΧΡΙ 8

    ΑΝ Ι < 5 ΤΟΤΕ

      ΟΜΙΛΟΣ[Ι] <-- 'Α'

      ΑΝ Ι = 1 ΤΟΤΕ

        ΓΡΑΨΕ 'ΔΩΣΕ ΟΝΟΜΑ ΟΜΑΔΑΣ ΚΑΙ ΒΑΘΜΟΥΣ ΠΡΩΤΑΘΛΗΜΑΤΟΣ'

        ΓΡΑΨΕ 'ΟΜΙΛΟΣ Α:'

      ΤΕΛΟΣ_ΑΝ

    ΑΛΛΙΩΣ

      ΟΜΙΛΟΣ[Ι] <-- 'Β'

      ΑΝ Ι = 5 ΤΟΤΕ

        ΓΡΑΨΕ 'ΔΩΣΕ ΟΝΟΜΑ ΟΜΑΔΑΣ ΚΑΙ ΒΑΘΜΟΥΣ ΠΡΩΤΑΘΛΗΜΑΤΟΣ'

        ΓΡΑΨΕ 'ΟΜΙΛΟΣ Β:'

      ΤΕΛΟΣ_ΑΝ

    ΤΕΛΟΣ_ΑΝ

    ΔΙΑΒΑΣΕ ΟΝ[Ι],ΒΑΘΜΟΙΠΡΩΤ[Ι]

    ΣΟΥΜΑ[Ι] <-- 0

  ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ

  ΓΙΑ Ι ΑΠΟ 1 ΜΕΧΡΙ 4

    ΔΙΑΦ1 <-- ΒΑΘΜΟΙΠΡΩΤ[Ι] - ΒΑΘΜΟΙΠΡΩΤ[4]

    ΒΑΘΜΟΙ[Ι] <-- ΑΡΧΙΚΗ(ΔΙΑΦ1)

    ΔΙΑΦ2 <-- ΒΑΘΜΟΙΠΡΩΤ[Ι+4] - ΒΑΘΜΟΙΠΡΩΤ[8]

    ΒΑΘΜΟΙ[Ι+4] <-- ΑΡΧΙΚΗ(ΔΙΑΦ2)

  ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ

  ΒΑΘΜΟΙ[1] <-- ΒΑΘΜΟΙ[1] + 3

  ΒΑΘΜΟΙ[5] <-- ΒΑΘΜΟΙ[5] + 3

  ΓΡΑΨΕ 'ΕΙΔΙΚΗ ΒΑΘΜΟΛΟΓΙΑ ΠΡΟ ΠΛΕΥ-ΟΦ ΑΝΑ ΟΜΙΛΟ:'

  ΓΡΑΨΕ 'ΟΜΙΛΟΣ Α:'

  ΓΙΑ Ι ΑΠΟ 1 ΜΕΧΡΙ 4

    ΓΡΑΨΕ ΟΝ[Ι], ΒΑΘΜΟΙ[Ι]

  ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ

  ΓΡΑΨΕ 'ΟΜΙΛΟΣ Β'

  ΓΙΑ Ι ΑΠΟ 5 ΜΕΧΡΙ 8

    ΓΡΑΨΕ ΟΝ[Ι], ΒΑΘΜΟΙ[Ι]

  ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ

  ΓΙΑ Ι ΑΠΟ 1 ΜΕΧΡΙ 8

    ΓΙΑ Κ ΑΠΟ 1 ΜΕΧΡΙ 8

      ΑΝ Ι <> Κ ΤΟΤΕ

        ΑΡΧΗ_ΕΠΑΝΑΛΗΨΗΣ

          ΓΡΑΨΕ 'ΔΩΣΕ ΤΟ ΑΠΟΤΕΛΕΣΜΑ ΤΟΥ ΑΓΩΝΑ', ΟΝ[Ι], 'ΕΝΑΝΤΙΟΝ', ΟΝ[Κ]

          ΔΙΑΒΑΣΕ ΑΓΩΝΕΣ[Ι,Κ]

        ΜΕΧΡΙΣ_ΟΤΟΥ ΑΓΩΝΕΣ[Ι,Κ] = 'Ν' Η ΑΓΩΝΕΣ[Ι,Κ] = 'Ι' Η ΑΓΩΝΕΣ[Ι,Κ] = 'Η'

      ΤΕΛΟΣ_ΑΝ

    ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ

  ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ

  ΚΑΛΕΣΕ ΥΠΟΛΟΓΙΣΜΟΣ (ΑΓΩΝΕΣ,ΣΟΥΜΑ)

  ΓΙΑ Ι ΑΠΟ 1 ΜΕΧΡΙ 8

    ΒΑΘΜΟΙ[Ι] <-- ΒΑΘΜΟΙ[Ι] + ΣΟΥΜΑ[Ι]

  ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ

  ΚΑΛΕΣΕ ΦΥΣΣΑΛΙΔΑ (ΒΑΘΜΟΙ, ΟΝ, ΟΜΙΛΟΣ)

  ΓΡΑΨΕ 'ΚΑΤΑΤΑΞΗ ΟΜΑΔΩΝ'

  ΓΙΑ Ι ΑΠΟ 1 ΜΕΧΡΙ 8

    ΓΡΑΨΕ Ι, ΟΝ[Ι], ΒΑΘΜΟΙ[Ι]

  ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ

  ΑΝ ΒΑΘΜΟΙ[2] > ΒΑΘΜΟΙ[3] ΤΟΤΕ

    ΓΡΑΨΕ 'ΟΙ ΟΜΑΔΕΣ', ΟΝ[1],'ΑΠΟ ΤΟΝ ΟΜΙΛΟ', ΟΜΙΛΟΣ[1], 'ΚΑΙ', ΟΝ[2],'ΑΠΟ ΤΟΝ ΟΜΙΛΟ', ΟΜΙΛΟΣ[2], 'ΑΝΕΒΑΙΝΟΥΝ ΑΠΕΥΘΕΙΑΣ'

    ΑΝ ΒΑΘΜΟΙ[3] > ΒΑΘΜΟΙ[4]  ΤΟΤΕ

      ΓΡΑΨΕ 'Η ΟΜΑΔΑ', ΟΝ[3], 'ΑΠΟ ΤΟΝ ΟΜΙΛΟ', ΟΜΙΛΟΣ[3], 'ΘΑ ΔΩΣΕΙ ΜΠΑΡΑΖ ΜΕ ΤΟΝ ΤΡΙΤΟ ΑΠ ΤΟ ΤΕΛΟΣ ΤΗΣ Α ΕΘΝΙΚΗΣ'

    ΑΛΛΙΩΣ

      ΔΕΙΚΤΗΣ <-- 3

      ΦΡΕΝΟ <-- ΨΕΥΔΗΣ

      ΟΣΟ ΒΑΘΜΟΙ[ΔΕΙΚΤΗΣ] = ΒΑΘΜΟΙ[ΔΕΙΚΤΗΣ + 1]  ΚΑΙ ΦΡΕΝΟ = ΨΕΥΔΗΣ ΕΠΑΝΑΛΑΒΕ

        ΔΕΙΚΤΗΣ <-- ΔΕΙΚΤΗΣ + 1

        ΑΝ ΔΕΙΚΤΗΣ = 8 ΤΟΤΕ

          ΦΡΕΝΟ <-- ΑΛΗΘΗΣ

        ΤΕΛΟΣ_ΑΝ

      ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ

      ΓΡΑΨΕ 'ΟΙ ΟΜΑΔΕΣ'

      ΓΙΑ Ι ΑΠΟ 3  ΜΕΧΡΙ  ΔΕΙΚΤΗΣ

        ΓΡΑΨΕ_ ΟΝ[Ι]

      ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ

      ΓΡΑΨΕ 'ΘΑ ΚΟΝΤΑΡΟΧΤΥΠΗΘΟΥΝ ΕΚ ΝΕΟΥ ΚΑΙ Ο ΠΡΩΤΑΘΛΗΤΗΣ ΘΑ ΔΩΣΕΙ ΑΓΩΝΕΣ ΜΠΑΡΑΖ'

    ΤΕΛΟΣ_ΑΝ

  ΑΛΛΙΩΣ_ΑΝ ΒΑΘΜΟΙ[1] > ΒΑΘΜΟΙ[2]  ΤΟΤΕ

    ΓΡΑΨΕ 'Η ΟΜΑΔΑ', ΟΝ[1], 'ΑΠΟ ΤΟΝ ΟΜΙΛΟ', ΟΜΙΛΟΣ[1], 'ΑΝΕΒΑΙΝΕΙ ΑΠΕΥΘΕΙΑΣ'

    ΔΕΙΚΤΗΣ <-- 2

    ΦΡΕΝΟ <-- ΨΕΥΔΗΣ

    ΟΣΟ ΒΑΘΜΟΙ[ΔΕΙΚΤΗΣ] = ΒΑΘΜΟΙ[ΔΕΙΚΤΗΣ + 1] ΚΑΙ ΦΡΕΝΟ = ΨΕΥΔΗΣ ΕΠΑΝΑΛΑΒΕ

      ΔΕΙΚΤΗΣ <-- ΔΕΙΚΤΗΣ + 1

      ΑΝ ΔΕΙΚΤΗΣ = 8 ΤΟΤΕ

        ΦΡΕΝΟ <-- ΑΛΗΘΗΣ

      ΤΕΛΟΣ_ΑΝ

    ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ

    ΓΡΑΨΕ 'ΟΙ ΟΜΑΔΕΣ'

    ΓΙΑ Ι ΑΠΟ 2  ΜΕΧΡΙ  ΔΕΙΚΤΗΣ

      ΓΡΑΨΕ_ ΟΝ[Ι]

    ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ

    ΓΡΑΨΕ 'ΘΑ ΚΟΝΤΑΡΟΧΤΥΠΗΘΟΥΝ ΕΚ ΝΕΟΥ. Ο ΠΡΩΤΑΘΛΗΤΗΣ Θ ΑΝΕΒΕΙ ΑΠΕΥΘΕΙΑΣ ΚΑΙ Ο ΔΕΥΤΕΡΟΣ ΘΑ ΔΩΣΕΙ ΑΓΩΝΕΣ ΜΠΑΡΑΖ'

  ΑΛΛΙΩΣ

    ΔΕΙΚΤΗΣ <-- 1

    ΦΡΕΝΟ <-- ΨΕΥΔΗΣ

    ΟΣΟ ΒΑΘΜΟΙ[ΔΕΙΚΤΗΣ] = ΒΑΘΜΟΙ[ΔΕΙΚΤΗΣ + 1] ΚΑΙ ΦΡΕΝΟ = ΨΕΥΔΗΣ ΕΠΑΝΑΛΑΒΕ

      ΔΕΙΚΤΗΣ <-- ΔΕΙΚΤΗΣ + 1

      ΑΝ ΔΕΙΚΤΗΣ = 8 ΤΟΤΕ

        ΦΡΕΝΟ <-- ΑΛΗΘΗΣ

      ΤΕΛΟΣ_ΑΝ

    ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ

    ΓΡΑΨΕ 'ΚΑΝΕΙΣ ΔΕΝ ΑΝΕΒΑΙΝΕΙ ΑΠΕΥΘΕΙΑΣ. ΟΙ ΟΜΑΔΕΣ'

    ΓΙΑ Ι ΑΠΟ 2  ΜΕΧΡΙ  ΔΕΙΚΤΗΣ

      ΓΡΑΨΕ_ ΟΝ[Ι]

    ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ

    ΓΡΑΨΕ 'ΘΑ ΚΟΝΤΑΡΟΧΤΥΠΗΘΟΥΝ ΕΚ ΝΕΟΥ. ΟΙ ΔΥΟ ΠΡΩΤΟΙ Θ ΑΝΕΒΟΥΝ ΑΠΕΥΘΕΙΑΣ ΚΑΙ Ο ΤΡΙΤΟΣ ΘΑ ΔΩΣΕΙ ΑΓΩΝΕΣ ΜΠΑΡΑΖ'

  ΤΕΛΟΣ_ΑΝ

ΤΕΛΟΣ_ΠΡΟΓΡΑΜΜΑΤΟΣ

ΣΥΝΑΡΤΗΣΗ  ΑΡΧΙΚΗ(Χ) : ΑΚΕΡΑΙΑ

ΜΕΤΑΒΛΗΤΕΣ

  ΑΚΕΡΑΙΕΣ: Χ, Υ, Ζ

ΑΡΧΗ

  Υ <-- Χ  DIV  4

  Ζ <-- Χ MOD  4

  ΑΝ Ζ >= 2 ΤΟΤΕ

    Υ <-- Υ + 1

  ΤΕΛΟΣ_ΑΝ

  ΑΡΧΙΚΗ <-- Υ

ΤΕΛΟΣ_ΣΥΝΑΡΤΗΣΗΣ

ΔΙΑΔΙΚΑΣΙΑ  ΥΠΟΛΟΓΙΣΜΟΣ(ΑΓ,ΒΑΘ)

ΜΕΤΑΒΛΗΤΕΣ

  ΑΚΕΡΑΙΕΣ: I, Κ, ΒΑΘ[8]

  ΧΑΡΑΚΤΗΡΕΣ: ΑΓ[8,8]

ΑΡΧΗ

  ΓΙΑ I ΑΠΟ 1 ΜΕΧΡΙ 8

    ΓΙΑ Κ ΑΠΟ 1 ΜΕΧΡΙ 8

      ΑΝ I<>Κ ΤΟΤΕ

        ΑΝ ΑΓ[I,Κ] = 'Ν' ΤΟΤΕ

          ΒΑΘ[I] <-- ΒΑΘ[I] + 3

        ΑΛΛΙΩΣ_ΑΝ ΑΓ[I,Κ] = 'Η' ΤΟΤΕ

          ΒΑΘ[Κ] <-- ΒΑΘ[Κ] + 3

        ΑΛΛΙΩΣ

          ΒΑΘ[I] <-- ΒΑΘ[I] + 1

          ΒΑΘ[Κ] <-- ΒΑΘ[Κ] + 1

        ΤΕΛΟΣ_ΑΝ

      ΤΕΛΟΣ_ΑΝ

    ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ

  ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ

ΤΕΛΟΣ_ΔΙΑΔΙΚΑΣΙΑΣ

ΔΙΑΔΙΚΑΣΙΑ  ΦΥΣΣΑΛΙΔΑ (ΤΑΞΙΝΟΜΗΣΗ, ΠΟΙΟΣ, ΓΚΡΟΥΠ)

ΜΕΤΑΒΛΗΤΕΣ

  ΑΚΕΡΑΙΕΣ: ΤΑΞΙΝΟΜΗΣΗ[8], Ι, Κ, ΤΕΜΠΑΚ

  ΧΑΡΑΚΤΗΡΕΣ: ΠΟΙΟΣ[8], ΓΚΡΟΥΠ[8], ΤΕΜΠΧΑΡ

ΑΡΧΗ

  ΓΙΑ Ι ΑΠΟ 2 ΜΕΧΡΙ 8

    ΓΙΑ Κ ΑΠΟ 8 ΜΕΧΡΙ Ι  ΜΕ ΒΗΜΑ  -1

      ΑΝ ΤΑΞΙΝΟΜΗΣΗ[Κ] > ΤΑΞΙΝΟΜΗΣΗ[Κ-1] ΤΟΤΕ

        ΤΕΜΠΑΚ <-- ΤΑΞΙΝΟΜΗΣΗ[Κ]

        ΤΑΞΙΝΟΜΗΣΗ[Κ] <-- ΤΑΞΙΝΟΜΗΣΗ[Κ-1]

        ΤΑΞΙΝΟΜΗΣΗ[Κ-1] <-- ΤΕΜΠΑΚ

        ΤΕΜΠΧΑΡ <-- ΠΟΙΟΣ[Κ]

        ΠΟΙΟΣ[Κ] <-- ΠΟΙΟΣ[Κ-1]

        ΠΟΙΟΣ[Κ-1] <-- ΤΕΜΠΧΑΡ

        ΑΝ ΓΚΡΟΥΠ[Κ] <> ΓΚΡΟΥΠ[Κ-1] ΤΟΤΕ

          ΤΕΜΠΧΑΡ <-- ΓΚΡΟΥΠ[Κ]

          ΓΚΡΟΥΠ[Κ] <-- ΓΚΡΟΥΠ[Κ-1]

          ΓΚΡΟΥΠ[Κ-1] <-- ΤΕΜΠΧΑΡ

        ΤΕΛΟΣ_ΑΝ

      ΤΕΛΟΣ_ΑΝ

    ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ

  ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ

ΤΕΛΟΣ_ΔΙΑΔΙΚΑΣΙΑΣ

